
Abstract: We study the notion of extensibility in functional data types, as a new
approach to the problem of decorating abstract syntax trees with additional infor-
mation. We observed the need for such extensibility while redesigning the data types
representing Haskell abstract syntax inside Glasgow Haskell Compiler (GHC).

Specifically, we describe a programming idiom that exploits type-level functions to
allow a particular form of extensibility. The approach scales to support existentials
and generalised algebraic data types, and we can use pattern synonyms to make it
convenient in practice.

1 Introduction

Back in the late 1970’s, David Turner’s inspirational work on SK-combinators

[Turner, 1979b,a], and his languages Sasl [Turner, 1976], KRC [Turner, 1982],

and Miranda [Turner, 1985], were hugely influential in the early development

of functional programming. They introduced a generation of young computer

scientists to the joy and beauty of functional programming, in a very direct and

concrete way: elegant ideas; simple, perspicuous writing; and compelling inter-

active implementations. David’s work has had sustained impact; for example,

Miranda had a major influence on the design of Haskell [Hudak et al., 2007].

Algebraic Data Types (ADTs) and pattern matching are now firmly estab-

lished as a core feature of any modern functional language. They first appeared

as a usable feature in Hope [Burstall et al., 1980], and were rapidly adopted

in ML [Milner, 1984], and in David Turner’s Miranda. ADTs make functional

languages a fertile ground in which to define and process tree-like structures.

However, trees often cannot grow ; once a data type is defined and compiled,

its definition cannot be extended by adding new data constructors, and/or by

adding new fields to its existing data constructors.

This lack of extensibility can be very painful. For example, at the centre of

all compilers stand tall trees representing the abstract syntax of terms. Com-

piler programs processing these trees often do so by decorating the trees with

additional information. For instance, a name resolution phase adds information

about names, and a type inference phase stores the inferred types in the rele-

vant nodes. We refer to such extra information as decorations. The additional

information may appear as new fields to the existing data constructors, and/or

new data constructors in data types representing the trees.

The compiler writer is then faced with two unpalatable choices. She can define

a new data type representing the output decorated tree, at the cost of much

duplication. Or she can write a single data type with all the necessary fields

and constructors, at the cost of having many unused fields and constructors at

different stages of compilation.

This dilemma is very real. The Glasgow Haskell Compiler (GHC) has a single

data type HsSyn that crosses several compiler phases; and a second entire data

type TH .Syntax for Template Haskell. Moreover, some Haskell libraries, notably

haskell-src-exts define yet another data type for Haskell source code. These

data types are large (dozens of types, hundreds of constructors) and are very

difficult to keep in sync.

In this paper we offer a systematic programming idiom that resolves the

dilemma, by providing a way to extend data types within Haskell. We leverage

type-level openness to allow extensibility of term-level data constructors.

Specifically, we make the following contributions

– We describe a simple but powerful programming idiom that allows a data

type to be extended both with extra constructor-specific fields and with

extra constructors (Section 3).

– We show that the idea can be extended to work for existentials and GADTs

(Section 3.10).

We discuss related work in Section 5 and conclude in Section 6.

On a personal note, David’s papers and language implementations played

a major role in drawing one of us (Simon) into the world of functional pro-

gramming. My very first paper, Yacc in Sasl [Peyton Jones, 1985], was a parser

generator for Sasl, and David acted as a mentor for me, at a time when I had

no idea what programming language research was, or how to do it. Thank you

David: I will be forever grateful for your encouragement and guidance in the

launch phase of my professional life.

2 The challenge

In this section, we demonstrate the problem of decorating trees, and sketch some

conventional ways to address it.

2.1 Tree-Decoration Problem

A compiler might need several variants of data types representing terms. For

example:

– We might want to label every node with its source location.

– After name resolution we might want to decorate names in the tree with

additional information, such as their namespace.

– After type inference we might want to decorate some (but not all) construc-

tors of the tree with inferred types.

– The type checker might record type abstractions and applications that are

not present in the source code. For this it would need to add new data

constructors to the type — and for these constructors a source location

might not make sense.

One approach is to declare a completely new data type for each variant, but

this is obviously unattractive because of the duplication it involves. In a real-

istic setting, the abstract syntax for a source language might have tens of data

types (expressions, patterns, guards, comprehensions, declarations, sequences,

bindings, matches, etc etc), and hundreds of data constructors altogether.

The Glasgow Haskell Compiler (GHC) makes an excellent (if incestuous)

case study for the challenge of extensibility. In GHC, the syntax of Haskell,

HsSyn, defines no fewer than 97 distinct data types with a total of 321 data

constructors. It would be completely infeasible to define multiple variants of

such a huge collection of types. Not only would it be terrible to duplicate the

data structures, but we would also have to duplicate general functions like the

pretty printer.

2.2 So what does GHC do?

Faced with this dilemma, what does GHC do in practice? It adopts a variety of

strategies:

– Straightforward parameterisation. The entire syntax is parameterised over

the type of variables, so that we have1

1 These types are much simplified, but they convey the right idea for present purposes.

parse :: String → HsExpr RdrName

rename :: HsExpr RdrName → HsExpr Name

typecheck :: HsExpr Name → HsExpr Id

For example, the type checker replaces each Name with an Id ; the latter is

a Name decorated with a Type.

– Extra data constructors. The data types include parts like

data HsPat id = . . .

| ConPat id [Located (HsPat id)]

| ConPatOut id . . . other fields . . .

where the type checker replaces all uses of ConPat with ConPatOut . This is

clearly unsatisfactory because the passes before the type checker will never

meet a ConPatOut but there is no static guarantee of that fact.

– Alternating data types. GHC needs to pin a source location on every source-

syntax node (e.g., for reporting errors). It does so by alternating between two

types. In the ConPat constructor above, the Located type is defined thus:

data Located x = L SrcLoc x

So there is a type-enforced alternation between HsPat and Located nodes.

This idiom works quite well, but is often tiresome when traversing a tree

because there are so many L nodes to skip over.

– Phase-indexed fields. GHC uses the power of type families (Chakravarty et al.

[2005]) to describe fields that are present only before or after a specific phase.

For example, we see

data HsExpr id = . . .

| ExplicitPArr (PostTc id Type) [LHsExpr id ]

where the PostTc type family is defined thus:

type family PostTc id a

type instance PostTc RdrName a = ()

type instance PostTc Name a = ()

type instance PostTc Id a = a

This idiom makes use of the fact that HsSyn is parameterised on the type

of identifiers, and that type makes a good proxy for the compiler phase. So

the first field of an ExplicitPArr is () after parsing and after renaming, but

is Type after type checking.

All this works well enough for GHC, but it is very GHC-specific. Other tools want

to parse and analyse Haskell source code define their own data types; the widely-

used library haskell-src-exts is a good example. Even GHC defines a com-

pletely separate data type for Template Haskell in Language.Haskell .TH .Syntax .

These data types rapidly get out of sync, and involve a great deal of duplicated

effort.

3 An idiom that supports data type extension

We now introduce a programming idiom that allows data types to be extended

in more systematic way than the ad-hoc tricks described above.

We explain our solution with a running example. This paper is typeset from

a literate Haskell source file using lhs2TeX [Hinze and Löh, 2015], and the code

runs on GHC 8.0 using a set of well-established language extensions.

{-# LANGUAGE TypeFamilies, DataKinds, ConstraintKinds #-}

{-# LANGUAGE GADTs, EmptyCase, StandaloneDeriving #-}

{-# LANGUAGE TypeOperators, PatternSynonyms #-}

{-# LANGUAGE FlexibleInstances, FlexibleContexts #-}

import GHC .Types (Constraint)

3.1 Extensible ADT Declarations

As a running example, consider the following language of simply-typed lambda

terms with integer literals, and explicit type annotations:

i ∈ integers

x , y ∈ variables

A,B ,C ∈ TYP ::= Int | A → B

L,M ,N ∈ EXP ::= i | x | M ::A | λx .N | L M

In Haskell, the language above can be declared as the following data types:

data Exp = Lit Integer

| Var Var

| Ann Exp Typ

| Abs Var Exp

| App Exp Exp

...

...

...

type Var = String

data Typ = Int

| Fun Typ Typ

The data type Exp is not extensible. Our idea is to make it extensible like this:

46 Najd S., Jones S.P.: Trees that Grow



data ExpX ξ = LitX (XLit ξ) Integer

| VarX (XVar ξ) Var

| AnnX (XAnn ξ) (ExpX ξ) Typ

| AbsX (XAbs ξ) Var (ExpX ξ)

| AppX (XApp ξ) (ExpX ξ) (ExpX ξ)

| ExpX (XExp ξ)

...

...

...

type family XLit ξ

type family XVar ξ

type family XAnn ξ

type family XAbs ξ

type family XApp ξ

type family XExp ξ

In this new data type declaration:

– ξ is a type index to ExpX . We call ξ the extension descriptor, because it

describes which extension is in use. For example ExpX TC might be a variant

of ExpX for the type checker for a language; we will see many examples

shortly.

– Each data constructor C has an extra field of type XC ξ, where XC is a type

family, or type-level function [Chakravarty et al., 2005]. We can use this field

to extend a data constructor with extra fields (Section 3.3). For example, if

we define XApp TC to be Typ, the App constructor of a tree of type ExpX TC

will have a Typ field.

– The data type has one extra data constructor ExpX , which has one field

of type XExp ξ. We can use this field to extend the data type with new

constructors (Section 3.4).

Now, we can use the above extensible data type to define a completely un-

decorated (UD) variant of ExpX as follows.

type ExpUD = ExpX UD

data UD

type instance XLit UD = Void

type instance XVar UD = Void

...

...

...

type instance XAnn UD = Void

type instance XAbs UD = Void

type instance XApp UD = Void

type instance XExp UD = Void

Since the non-decorated variant does not introduce any forms of extensions,

all mappings are set to Void2 which is declared (in Data.Void) like this:

data Void

void :: Void

void = error "Attempt to evaluate void"

...

...

...

absurd :: Void → a

absurd m = case m of { }

That is, Void is a data type with no constructors, so it is inhabited only by

bottom.

2 ignoring the bottom type, () is used for empty field decorations, since for products
(constructor fields) types ((),A) and (A, ()) are isomorphic to A

47Najd S., Jones S.P.: Trees that Grow



With this instantiation, ExpX UD is almost exactly3 isomorphic to the orig-

inal data type Exp; that is, there is a 1-1 correspondence between values of

ExpX UD and values of Exp.

The alert reader may realise that the type instance declarations can all be

omitted because, in the absence of such instances, XAnn UD is irreducible and

hence is an empty type just like Void . But then there is no way to prevent clients

of ExpX UD from accidentally adding an instance for XAnn UD, so we generally

prefer to prevent that by giving an explicit instance.

3.2 Pattern Synonyms for Convenience

One can program directly over the new ExpX type, but it is a bit less convenient

than it was with the original Exp data type:

– When pattern matching, we must ignore the extra field in each constructor.

– When constructing, we must supply void in the extra field.

For example:

incLit :: Exp → Exp

incLit (Lit i) = Lit (i + 1)

incLit e = e

incLitX :: ExpUD → ExpUD

incLitX (LitX i) = LitX void (i + 1) -- Tiresome clutter

incLitX e = e

Solving this kind of inconvenience is exactly what pattern synonyms were

invented for [Pickering et al., 2016]. We may define a pattern synonym thus

pattern LitUD :: Integer → ExpUD

pattern LitUD i ← LitX i

where LitUD i = LitX void i

and similarly for all the other data constructors. This is a so-called bidirectional

pattern synonym. In a pattern LitUD i expands to LitX i , while in a term

LitUD i expands to LitX void i . So now we can write

incLitX :: ExpUD → ExpUD

incLitX (LitUD i) = LitUD (i + 1) -- No tiresome clutter

incLitX e = e

3 We say “almost exactly” because the term value ExpX void has no counterpart
in Exp; alas Haskell lacks an entirely uninhabited type. We can simply hide the
constructor ExpX from the client users to ameliorate this problem.

3.3 New Field Extensions

Now, consider the following simple type system for our example language.

Γ �M : A
Γ � i : Int

(x : A) ∈ Γ

Γ � x : A
Γ � M : A

Γ � M ::A : A

x : A,Γ � N : B

Γ � λx .N : A → B
Γ � L : A → B Γ � M : A

Γ � L M : B

Before type checking, often abstract syntax trees (ASTs) are processed by a

type inference engine. The output of the type inference engine is the same input

tree decorated with additional type information. Type inference helps users to

leave certain bits of their programs without explicit type annotations. Type

inference also helps in simplifying the type checker: after type inference, and

decorating the trees with the additional type information, type checking becomes

a straightforward syntax-directed recursive definition. To accommodate for the

additional information in the output, we need larger trees, and hence we need

to extend the original declarations. For instance, the following highlights the

required changes to the non-extensible Exp data type:

data Exp = . . . | App Typ Exp Exp

The definition is just like that of Exp, save for extending constructor AppX
with a new field of type Typ, as highlighted above. The duplication is unpleasant

(particularly when the data type is much larger).

In the extensible setting both non-decorated and decorated variants of Exp

can be defined as extensions to the same base extensible data type ExpX . Fol-

lowing the same approach as before, we can also define a decorated variant of

ExpX suitable for type checking (TC) based on ExpX as follows.

type ExpTC = ExpX TC

data TC

type instance XLit TC = Void

type instance XVar TC = Void

...

...

...

type instance XAnn TC = Void

type instance XAbs TC = Void

type instance XApp TC = Typ

type instance XExp TC = Void

The difference (highlighted) is just that the App constructor gets an extra

field of type Typ, just as required.

The pattern synonyms for ExpTC can be defined as before, save for con-

structor AppX that takes an extra argument for the new field introduced by the

extension, as highlighted below:

pattern AppTC :: Typ → ExpTC → ExpTC → ExpTC

pattern AppTC a l m = AppX a l m

3.4 New Constructor Extensions

We could as well consider scenarios where extended data types introduce new

constructors. For instance, consider a simple partial evaluation (PE) pass over

the trees, where β-redices are normalised away in an input tree. After reducing a

redex to a value, the partial evaluator stores the value as a relevant node inside

the tree. To be able to decorate the tree with this new information (i.e., values),

often new constructors should be introduced to the declarations. For instance,

the following highlights the required changes (i.e., the new constructor Val) to

the non-extensible Exp data type:

data Val = . . .

data Exp = . . . | Val Val

We can still reuse our extensible data type ExpX to define a variant suitable for

such partial evaluation (PE) by extending it with a new constructor ValPE as

type ExpPE = ExpX PE

data PE

type instance XLit PE = Void

type instance XVar PE = Void

...

...

...

type instance XAnn PE = Void

type instance XAbs PE = Void

type instance XApp PE = Void

type instance XExp PE = Val

The pattern synonyms for ExpX PE can be defined as before, except that

we introduce a new pattern synonym ValPE that represents the new constructor

introduced by the extension, as highlighted below:

pattern ValPE :: Val → ExpPE

pattern ValPE v = ExpX v

3.5 Normal Functions on Extended Data Types

Aided by the pattern synonyms, programming over the extended data type feels

very much like programming over an ordinary non-extensible data type. For

example, here is a type checker following the typing rules in Section 3.3:

check :: ExpTC → [(Var ,Typ)]→ Typ → Bool

check (LitTC ) Int = True

check (VarTC x ) Γ c = maybe False (≡ c) (lookup x Γ )

check (AnnTC m a) Γ c = a ≡ c ∧ check m Γ c

check (AbsTC x n) Γ (Fun a b) = check n ((x , a) : Γ ) b

check (AppTC a l m) Γ c = check l Γ (Fun a c) ∧ check m Γ a

check = False

One significant annoyance is that GHC is not yet clever enough to know when

pattern synonyms are exhaustive, so the pattern-match exhaustiveness checker

is effectively powerless.

3.6 Generic Functions on Extensible Data Types

We can sometimes exploit the common structure inherited from an extensible

data type, to define generic functions acting uniformly over the extending data

types. For instance, we can define a generic printer function once and reuse it

for the extended data types. Let us begin with a simple printer that ignores

the decorations introduced as new fields in the data type. For instance, such

a printer works the same for both the undecorated data type ExpUD and the

decorated data type ExpTC . Compilers often use such printers across multiple

phases to print terms while reporting error messages.

For the new constructor extensions, we can either ignore them like we do

for the new fields, or use function parameters to handle printing of these new

constructors. We choose to do the latter in the following example.

printT :: Typ → String

printT Int = "Int"

printT (Fun a b) = "("++ printT a ++ ") → "++ printT b

printE :: (XExp ξ → String)→ ExpX ξ → String

printE (LitX i) = show i

printE (VarX x ) = x

printE p (AnnX m a) = "("++ printE p m ++ ")::("++ printT a ++ ")"

printE p (AbsX x n) = "λ"++ x ++ "."++ printE p n

printE p (AppX l m) = "("++ printE p l ++ ") ("++ printE p m ++ ")"

printE p (ExpX ξ) = p ξ

Above, we chose to pass explicitly the function parameters used for printing the

possible new constructors. We could as well use type classes.

Having defined the above generic printer, we can reuse it to define printers

for extending data types ExpUD, ExpTC , and ExpPE as follows.

printEUD :: ExpUD → String

printEUD = printE absurd

printETC :: ExpTC → String

printETC = printE absurd

...

...

...

printEPE :: ExpPE → String

printEPE = printE p

where p v = "{{"++ show v ++ "}}"

deriving instance Show Val

Since both ExpUD and ExpTC introduce no new constructors, the parameters

passed to the generic function does plain matching on empty types. For ExpPE ,

however, we pass a printer function handling values of the new constructor Val .

3.7 Type Classes for Extensible Data Types

For the generic printer printE , we chose to ignore the new field extensions. We

could as well make a variant that also prints the new field extensions. Such a

printer is useful for debugging purposes. To implement such a printer for ExpX ,

as before, we need to provide five more function parameters to handle new field

extensions in each of the five constructors. The type of generic function then

becomes

printE :: (XLit ξ → String)→ (XVar ξ → String)→ (XAnn ξ → String)→

(XAbs ξ → String)→ (XApp ξ → String)→ (XExp ξ → String)→

ExpX ξ → String

Here, with this approach, genericity comes at the price of a long list of pa-

rameters that need to be passed around. But this is exactly what Haskell’s type

classes were designed to solve! We can instead write4

instance (Show (XLit ξ),Show (XVar ξ),Show (XAnn ξ),

Show (XAbs ξ),Show (XApp ξ),Show (XExp ξ))⇒

Show (ExpX ξ) where

show = ...

and all the extra parameter passing becomes invisible.

Using the constraint kinds extension in GHC, we can make the process of

declaring such generic instances easier by abstracting over the constraint as

type ForallX (φ :: ∗ → Constraint) ξ

= (φ (XLit ξ), φ (XVar ξ), φ (XAnn ξ)

, φ (XAbs ξ), φ (XApp ξ), φ (XExp ξ))

Hence by using above, the header of the instance declaration for ExpX be-

comes as simple as

instance ForallX Show ξ ⇒ Show (ExpX ξ) where

show = ...

In this case, and many others, we can even use Haskell’s automatic (standalone)

instance deriving to implement the show method for us:

deriving instance ForallX Show ξ ⇒ Show (ExpX ξ)

4 The type checker complains about the decidablity of type class resolution, because
the constraints in the instance context are no smaller than those in the head. There-
fore, we need to supply the compiler with a −XUndecidableInstances flag, because
we, as the programmers, know that the process is terminating for our use cases.

3.8 Replacing Constructors

In some compiler passes, changes to trees can be beyond mere extensions. For

instance, a pass may require the type of a field in a constructor to change.

Consider the common pass in compilers where a chain of term applications are

grouped together to form a saturated application where the term at the head

of the chain is directly applied to a list of terms (often with η-expansions so

that the size of the arguments list matches the arity of the head term). In our

running example, to store the result of the saturation pass in a variant of Exp,

we change the type of the arguments in constructor App to a list of terms:

data Exp = . . . | App Exp [Exp ]

Such a change to the type of a field in a constructor, and in general changes

to a constructor beyond what can be achieved by adding new fields (and smart

use of pattern synonyms) can still be achieved following our idiom by replacing

the constructor with a new one.

The act of replacing a constructor can be seen as two distinct extensions:

(a) adding a new constructor, (b) removing the old constructor. Removing a

constructor is achieved by extending it with a new field of an empty type. As

mentioned earlier, Haskell does not have such an empty type, as all types are

inhabited by bottom, but we can achieve a similar result by not exposing the

removed constructor to the client user (as a part of the interface of the extended

data type).

Assuming AppX is not exposed to the client users, the following defines a vari-

ant of ExpX with fully saturated applications (SA), where the type of arguments

in application terms is changed to a list of terms .

type ExpSA = ExpX SA

data SA

type instance XLit SA = Void

type instance XVar SA = Void

...

...

...

type instance XAnn SA = Void

type instance XAbs SA = Void

type instance XApp SA = Void

type instance XExp SA

= (ExpSA, [ExpSA ])

Now the new exposed application constructor can be defined by the following

pattern synonym:

pattern AppSA :: ExpSA → [ExpSA ]→ ExpSA

pattern AppSA l ms = ExpX (l ,ms)

3.9 Extensions Using Type Parameters

The running example we considered so far has had no type parameters, besides

the extension descriptor parameter that we have introduced. Many of the data

types that we want to extend do have type parameters. For instance, consider a

variant of Exp that is parametric at the type of variables:

data Exp α = . . . | Var α | Abs α (Exp α)

Also consider a variant of the above with additional let expressions, as often

introduced by passes such as let-insertion:

data Exp α = . . . | Let α (Exp α) (Exp α)

Our idiom can also describe such an extension even though the extension (i.e.,

type variables in let bindings) is referring to the type parameter α. The general

idea is to directly pass the type parameters in a constructor to the corresponding

extension type functions:

data ExpX ξ α

= LitX (XLit ξ α ) Integer

| VarX (XVar ξ α ) Var

| AnnX (XAnn ξ α ) (ExpX ξ α ) Typ

| AbsX (XAbs ξ α ) Var (ExpX ξ α )

| AppX (XApp ξ α ) (ExpX ξ α ) (ExpX ξ α )

| ExpX (XExp ξ α )

...

...

...

type family XLit ξ α

type family XVar ξ α

type family XAnn ξ α

type family XAbs ξ α

type family XApp ξ α

type family XExp ξ α

The extension introducing let expressions (LE) is defined same as before,

this time with access to the type parameter:

type ExpLE α = ExpX LE α

data LE

type instance XLit LE α = Void

type instance XVar LE α = Void

...

...

...

type instance XAnn LE α = Void

type instance XAbs LE α = Void

type instance XApp LE α = Void

type instance XExp LE α

= (α,ExpLE α,ExpLE α)

Now, we can define a pattern synonym for the new constructor as before:

pattern LetLE :: α→ ExpLE α→ ExpLE α→ ExpLE α

pattern LetLE x m n = ExpX (x ,m,n)

Similarly, we can support extensible data types with more than one type

variable by passing them all to the extension type functions.

3.10 Existentials and GADTs

So far, we have considered extensibility in normal algebraic data type declara-

tions. However, in GHC, data may be defined by generalised algebraic data type

(GADT) declarations. For instance, consider the following GADT declaration of

a simple embedded domain-specific language of constants, application, addition,

and Boolean conjunction, with one existential variable a in App:

data Exp a where

Con :: c → Exp c

App :: Exp (a → b)→ Exp a → Exp b

Add :: Exp (Int → Int → Int)

And :: Exp (Bool → Bool → Bool)

We cannot print terms of this type: due to the polymorphic type of the field

in Con, we need a printer for values of type α when printing Exp α, and since

a is locally quantified in App and unavailable outside, we cannot supply such a

printer from outside. We need to extend constructor App to store a printer for

a right inside the constructor:

data Exp a where

App :: (a → String) → Exp (a → b)→ Exp a → Exp b

. . .

Our idiom scales to generalised algebraic data types and supports extensibil-

ity as above. To do so, we need to be able to access existential variables when

defining extensions. As in the previous section, we can do so by simply pass-

ing the existential types to the extension type functions as well. For the above

example, we have the following extensible declaration:

data ExpX ξ a where

ConX :: XCon ξ c → c → ExpX ξ c

AppX :: XApp ξ a b → ExpX ξ (a → b)→ ExpX ξ a → ExpX ξ b

AddX :: XAdd ξ → ExpX ξ (Int → Int → Int)

AndX :: XAnd ξ → ExpX ξ (Bool → Bool → Bool)

ExpX :: XExp ξ a → ExpX ξ a

type family XCon ξ c

type family XApp ξ a b

type family XAdd ξ

...

...

type family XAnd ξ

type family XExp ξ a

We can now define a variant of ExpX , where AppX is extended with a new

field to store a printer (Pr) for the existential type a:

type ExpPr a = ExpX Pr a

data Pr

type instance XCon Pr c = Void

type instance XApp Pr a b = (a → String)

...
:

type instance XAdd Pr = Void

type instance XAnd Pr = Void

type instance XExp Pr a = Void

As before, we can define pattern synonyms such as the following:

pattern AppPr :: (a → String)→ ExpPr (a → b)→ ExpPr a → ExpPr b

pattern AppPr p l m = AppX p l m

One other solution for writing such a printer is to constrain the indices of

Exp with Show type class. This involves adding a local type constraint for the

existential type a in AppE . Our idiom is also capable of expressing such exten-

sions to the set of local type constraints. For this purpose, we need to introduce

a proof data type Proof φ a that matching on its constructor convinces GHC

that the constraint φ a is satisfied. So to define a variant of ExpX where the

existential type is constrained with Show type class (Sh), we have the following.

data Proof φ a where

Proof :: φ a ⇒ Proof φ a

type ExpSh a = ExpX Sh a

data Sh

type instance XCon Sh c = Void

type instance XApp Sh a b = Proof Show a

...

...

type instance XAdd Sh = Void

type instance XAnd Sh = Void

type instance XExp Sh a = Void

Now, we can define pattern synonyms such as following:

pattern AppSh :: ()⇒ Show a ⇒ ExpSh (a → b)→ ExpSh a → ExpSh b

pattern AppSh l m = AppX Proof l m

Similarly, we can add new locally quantified variables using a data type like

data Exists f where

Exists :: f a → Exists f

3.11 Variations on Theme

Our idiom is just that: a programming idiom. For field extensions (Section 3.3),

nothing requires us to add an extra field to every constructor, or to use a different

type function for every constructor. Similarly if we do not want to extend the

data type with new constructors we do not need to provide the extra data

constructor that supports such extension (Section 3.4). For example, here is a

more specialised variant of our running example

data ExpX ξ = LitX (XLeaf ξ) Integer

| VarX (XLeaf ξ) Var

| AnnX (ExpX ξ) Typ

| AbsX (XAA ξ) Var (ExpX ξ)

| AppX (XAA ξ) (ExpX ξ) (ExpX ξ)

...

...

...

type family XLeaf ξ

type family XAA ξ

Here constructors LitX and VarX share a single extension-field type, XLeaf ξ;

and similarly AbsX and AppX ; the constructor AnnX does not have an extension

field; and we cannot add new data constructors.

3.12 Shortcomings and Scope

Our approach comes with a number of shortcomings.

– Efficiency. Every constructor carries an extra extension field, whether or not

it is used.

– Exhaustiveness checks. Our use of pattern synonyms (which is optional, of

course) defeats GHC’s current pattern-match exhaustiveness checker. And

even if we did not use a pattern synonym, the extra constructor (ExpX in

our running example) will be flagged as unmatched even when we are not

using it. Both are problems of engineering, rather than fundamental. 5

– Boilerplate. When adding a new phase descriptor, there is a slightly uneasy

choice between (a) adding lots of tiresome declarations

type instance XC ξ = Void

one for each constructor C whose extension field is not used, and (b) omitting

the instance, and hoping that no one adds it later.

Similarly, writing lots of pattern-synonym declarations can be painful.

One alternative we have considered is to generate the boilerplate using Tem-

plate Haskell, or even to define a new language extension. But it seems better

first to gain more experience of using the idiom.

Our idiom can naturally scale to support mutually recursive declarations by

passing the same extension descriptor to all of the declarations.

We have seen that our idiom is capable of expressing extensions to a gener-

alised algebraic data type declaration such as adding new fields, adding new

5 In fact, there are already partially implemented general features in GHC regarding
both completeness of a set of pattern synonyms, and improving the totality checker
to recognise absurdity.

constructors, adding new local constraints, and adding new existential variables.

We have also seen that, we can replace constructors, and access global and local

type variables in our extensions.

In addition to these changes, we can combine our idiom with pattern syn-

onyms and module system features to express other changes like

– change to the order of fields, such as

pattern (
) :: Exp → Exp → Exp

pattern m 
 l = App l m

– removing fields, such as

pattern K :: Exp → Exp

pattern K n ← Abs n

where K n = Abs "_" n

– fixing values of fields, such as

pattern One :: Exp

pattern One = Lit 1

Yet, there are other possible forms of changes to a data type declaration, like

adding new type variables. In the next section, we take a few steps further.

4 Extension Descriptors

So far in our examples, the extension description parameters have been empty

types used as indices to define extensions. However, extension descriptors are

themselves ordinary algebraic data types, and in this section we study extensi-

bility using more complex extension descriptors.

4.1 New Type Parameter Extensions

Suppose we wanted to add a new field of type α to some or all of the data

constructors in the type. Then we would need to add α as a parameter of the

data type itself. Can we do that?

In our example, suppose we wanted to add a source location to every node

in the tree. Source location decorations associated with a node may appear

as new fields, or as new constructors wrapping nodes with a source location.

Strictly speaking, the latter approach is less precise compared to the former:

such wrapper constructors can be applied to a node more than once, or not ap-

plied at all. With the former, the programmer is in control: using the optional

type (e.g., Maybe) of source locations in decorations models the optional appli-

cation of wrapper constructors, and using the list of source locations models the

multiple applications of wrapper constructors to a node. Regardless of the dec-

oration approach, the type of source locations (annotations in general) is often

kept polymorphic, allowing programmers to define generic functions like fmap,

fold , and traverse. A good example is the AST in Haskell-Src-Exts, where the

polymorphic annotations are used for different purposes, including the source

locations used in an exact printer. In our extensible setting, support for poly-

morphic source locations amounts to (1) extending the AST declarations with

a new type parameter α (the type of source locations) and (2) extending all the

constructors with a new field of type α. To do so, we need the ability to extend

an ADT data type declaration with a set of type variables, and to access these

variables to define extensions, such as new fields. Our encoding is capable of ex-

pressing such new type parameter extensions: the idea is to carry the extra type

parameters in the extension descriptors. For instance, the following defines an

extension to ExpX with a new type variable α, and uses it to define polymorphic

annotations An as new field extensions.

type ExpAn α = ExpX (An α )

data An α

type instance XLit (An α ) = α

type instance XVar (An α ) = α

...

...

...

type instance XAnn (An α ) = α

type instance XAbs (An α ) = α

type instance XApp (An α ) = α

type instance XExp (An α ) = Void

Notice that we made the definition of the extension descriptor parametric,

and then we could access the parameter when defining extensions.

4.2 Hierarchy of Extension Descriptors

In practice, compilers may have multiple variants of an AST, many of which are

closely related to each other. For instance in GHC, the AST in the front-end

of the compiler, named HsSyn, has three major variations used in the parsing,

renaming, and type-checking passes. GHC also has an entirely separate variant as

a part of its metaprogramming mechanism Template Haskell. The first three are

closely related, while the last is quite different. We can organise such variations

by putting them in hierarchies of indices and use this hierarchy when defining

extensions. For instance, for GHC, we may define the extension descriptor as

data GHC (c :: Component)

data Component = Compiler Pass | TemplateHaskell

data Pass = Parser | Renamer | TypeChecker
Having the above as a hierarchy of extension descriptors, we get the four vari-

ations of HsSyn AST in the extensible setting. For instance, the type checker

AST would be of the type HsSyn (Compiler TypeChecker).

It also allows us to define generic extension descriptors such as

type family PostTC p where

PostTC TypeChecker = Typ

PostTC = Void

type instance XApp (GHC TemplateHaskell) = Void

type instance XApp (GHC (Compiler p)) = PostTC p

5 Discussion and Related Works

The problem of extensibility in data types is a hot topic in computer science.

There are many different approach to this problem. To name a few: struc-

tural and nominal subtyping, extensible records and variants, and numerous

approaches to Wadler’s expression problem. There are too many solutions to

mention; the reader may consult the references in [Torgersen, 2004; Axelsson,

2012; Swierstra, 2008; Lindley and Cheney, 2012; Bahr and Hvitved, 2011; Löh

and Hinze, 2006], for some examples. However, surprisingly, our problem, and

hence our solution, has unique distinguishing characteristics:

Need for both directions of data extensibility: We need, and provide, ex-

tensibility on two major directions of data extensibility: adding new fields to

existing constructors, and adding new constructors. The so-called “expres-

sion problem” and its solutions are often only concerned with the latter form

of extensibility.

Generic programming is a plus, not a must: Our primary goal is to re-

use the data type declaration itself, rather than to re-use existing functions

that operate over that type. For example, in GHC, the parser produces

HsSyn RdrName, the renamer consumes that and produces HsSyn Name,

which the type checker consumes to produce HsSyn Id . All three passes

are monomorphic: they consume and produce a single fixed variant of the

underlying data type.

In contrast, work addressing the expression problem is often concerned with

re-usability and compositionality of functions defined per cases.

As we have seen with some examples (e.g., the generic printer), one can

write and reuse functions that are polymorphic in the extension descriptor,

but only by (a) simply discarding or preserving the decorations, or (b) using

auxiliary higher order functions to process the decorations. If one wishes to

take functions written only for a specific variant of a data type and reuse

them, as an after-thought, for other variants, certain forms of static guaran-

tees (possibly, beyond what types currently provide) are required for safety.

One common practice here is to focus on certain subclass of data types.

Trees are declared: In our setting, trees are often declared, rather than them

being anonymous. There are well-known trade-offs between declared and

anonymous data structures. The former is simpler and less error-prone, and

the latter enables more opportunities for generic programming. Row poly-

morphism, and the similar, often infer the structure of data from their uses,

leading to large types, bad performance, and complicated error messages.

Our approach is based on declaring both extensible and extended data types

(by describing the exact extensions). It resembles the long lasting problem of

supporting anonymous records, such as Trex [Gaster, 1998], in GHC, where

solutions with declared flavour often dodge the problem by leaving program-

mers to do some of the work by providing more information.

Similar to our idiom in spirit is McBride’s Ornaments [Dagand and McBride,

2014; Williams et al., 2014]. The key idea of ornaments is to declare trans-

formations of data types that preserve the recursive structures of data types,

with focus on reusing functions defined on the original for the transformed

data types. While our idiom can benefit from works on ornaments for such

reuse, there are decorations in practice that do not preserve the recursive

structures. For instance, in GHC, for better or worse, the constructor rep-

resenting if-expressions (like some others) is decorated with one additional

expression to store user-defined macros rebinding if-syntax, hence not pre-

serving the recursive structure.

Works with the current technology: Existing solutions often demand

changes to the compiler. Some other, come at the price of losing certain

desirable properties, such as decidablity of type inference, or predictability

of the performance. In contrast, our solution works in GHC right now (v8.0).

6 Conclusion

In the 1980s we were mainly concerned with functional programming over terms,

but this paper has mainly focused on functional programming over types, with

the interesting new twist that type functions (unlike term functions) can be

open. We have explored how to leverage that type-level openness to allow exten-

sibility of term-level data constructors. David, we hope that you approve. Happy

birthday!

